
! Lagerexempel med hemlig bakgrund med kompletteringar av Peter Lohmander 101107;
! Viss uppgradering 121102;

model:

sets:
time/1..7/:IndLag,VagLag,Trp,Prod,Forb,Ext;
endsets

! time 1 representerar April;
! Lagernivaaer avser ingaende lagernivaaer;

Max = NUV;

NUV = @SUM(time(t)| t #LT#7: @EXP(-r*t)*(
                                          - 90 * Prod(t)    
                                          - 10 * VagLag(t) 
                                          - 15 * IndLag(t) 
                                          - 50 * Trp(t)    
                                          - 400 * Ext(t) 
                                          + 800 * Forb(t)
                                         )
          );

[rate]r = .05/12;

[Vag0]VagLag(1) = 3000;
                 
[Ind0]IndLag(1) = 3000;

[SumProd]@SUM(time(t)| t #LT#7: Prod(t)) = 14000; 

Prod(7) = 0;

@for(time(t)| t #LT#7: [VagBal] VagLag(t) + Prod(t) - Trp(t) - VagLag(t+1) = 0) ;

@for(time(t)| t #LT#7: [IndBal] IndLag(t) + Trp(t) + Ext(t) - Forb(t) - IndLag(t+1) = 0) ;

@for(time(t): [ILAGMAX]IndLag(t) <= 4000);
@for(time(t): [ILAGMIN]IndLag(t) >= 500);

!@for(time(t)| t #LT#7: [VLAGMAX] VagLag(t) <= 0.6 * Trp(t+1)); 
@for(time(t)| t #LT#7: [VLAGMIN] VagLag(t) >= 10); 

@for(time(t)| t #LT#7: [TrpMAX] Trp(t) <= 3000); 
@for(time(t)| t #LT#7: [TrpMIN] Trp(t) >= 1500); 

@for(time(t)| t #GT#3: [Extlim] Ext(t) = 0); 

data:
Forb =     4000 3000 3000 4000 3000 3000 3000;
@OLE('LT121102.XLS')= VagLag,IndLag, Trp, Prod, Forb, Ext, Nuv;
enddata
end




